

ΚΕΦΑΛΑΙΟ 10

Στο κεφάλαιο αυτό θα εφαρμόσουμε μεθόδους πινάκων για τη λύση του κβαντομηχανικού προβλήματος του προσδιορισμού των ενεργειακών επιπέδων του αναρμονικού ταλαντωτή. Το πρόβλημα αυτό δεν μπορεί να λυθεί ακριβώς αναλυτικά οπότε πρέπει να προσφύγουμε σε διαταρακτικές ή αριθμητικές μεθόδους. Εμείς θα αντιμετωπίσουμε το πρόβλημα αριθμητικά. Για το σκοπό αυτό θα προσδιορίσουμε κατάλληλη βάση για να εκφράσουμε τη Χαμιλτονιανή H υπό μορφή πίνακα και θα τον διαγωνιοποιήσουμε αριθμητικά. Φυσικά ο πίνακας αυτός είναι απείρου μεγέθους και θα χρησιμοποιήσουμε την προσέγγιση η βάση που θα επιλέξουμε να έχει πεπερασμένο αριθμό από μέλη. Για να ελέγξουμε την ακρίβεια των αποτελεσμάτων μας, θα πρέπει να βεβαιωθούμε πως οι υπολογιζόμενες τιμές συγκλίνουν με την επιθυμητή ακρίβεια στις πραγματικές τιμές καθώς θα υπολογίζουμε με συνεχώς αυξανόμενο μέγεθος της βάσης στο χώρο Hilbert.

Για τον υπολογισμό των ιδιοτιμών θα χρησιμοποιήσουμε τις υπορουτίνες που βρίσκουμε στην βιβλιοθήκη LAPACK, θα είναι λοιπόν μια άσκηση για το πώς να συνδέουμε το πρόγραμμά μας με υπάρχουσες βιβλιοθήκες.

Το κεφάλαιο αυτό βασίζεται στα Mathematica Notebooks του Peter West (<http://bartok.ucsc.edu/peter/115>) και στις σημειώσεις “Computational Physics” των U. Wolff, B. Bunk και F. Knechtli. Ο φοιτητής που ενδιαφέρεται μπορεί να ανατρέξει σε αυτές και να δει πώς λύνονται τα προβλήματα με τη χρήση Mathematica και Matlab αντίστοιχα.

10.1 ΕΙΣΑΓΩΓΗ

Η Χαμιλτονιανή του αρμονικού ταλαντωτή δίνεται από τη σχέση:

$$H_0 = \frac{p^2}{2m} + \frac{1}{2}m\omega^2 x^2 \quad (10.1)$$

και ορίζοντας $x_0 = \sqrt{\hbar/(m\omega)}$, $p_0 = \sqrt{\hbar m\omega}$ παίρνουμε την εξίσωση αδιάστατων μεγεθών:

$$\frac{H_0}{\hbar\omega} = \frac{1}{2}\left(\frac{p}{p_0}\right)^2 + \frac{1}{2}\left(\frac{x}{x_0}\right)^2. \quad (10.2)$$

Μετρώντας την ενέργεια σε μονάδες $\hbar\omega$, τις αποστάσεις σε μονάδες x_0 και τις ορμές σε μονάδες p_0 παίρνουμε

$$H_0 = \frac{1}{2}p^2 + \frac{1}{2}x^2 \quad (10.3)$$

Ο τελεστής H_0 μπορεί να διαγωνιοποιηθεί εύκολα με τη βοήθεια των τελεστών δημιουργίας/καταστροφής:

$$x = \frac{1}{\sqrt{2}}(a^\dagger + a) \quad p = \frac{i}{\sqrt{2}}(a^\dagger - a) \quad (10.4)$$

ή

$$a = \frac{1}{\sqrt{2}}(x + ip) \quad a^\dagger = \frac{1}{\sqrt{2}}(x - ip) \quad (10.5)$$

που ικανοποιούν τη σχέση μετάθεσης

$$[a, a^\dagger] = 1 \quad (10.6)$$

και τότε

$$H_0 = a^\dagger a + \frac{1}{2}. \quad (10.7)$$

Οι ιδιοκαταστάσεις της H_0 ικανοποιούν τις σχέσεις ($n = 0, 1, 2, \dots$)

$$a^\dagger |n\rangle = \sqrt{n+1} |n+1\rangle \quad a |n\rangle = \sqrt{n} |n-1\rangle \quad a |0\rangle = 0 \quad (10.8)$$

οπότε

$$a^\dagger a |n\rangle = n |n\rangle \quad (10.9)$$

και

$$H_0 |n\rangle = E_n |n\rangle, \quad E_n = n + \frac{1}{2}. \quad (10.10)$$

Η αναπαράσταση θέσης των ιδιοκαταστάσεων $|n\rangle$ είναι:

$$\psi_n(x) = \langle x | n \rangle = \frac{1}{\sqrt{2^n n! \sqrt{\pi}}} e^{-x^2/2} H_n(x) \quad (10.11)$$

όπου $H_n(x)$ τα πολώνυμα Hermite.

Από τις σχέσεις (10.4), (10.8) προκύπτει ότι

$$x_{nm} = \langle n | x | m \rangle = \frac{1}{\sqrt{2}} \sqrt{m+1} \delta_{n,m+1} + \frac{1}{\sqrt{2}} \sqrt{m} \delta_{n,m-1} \quad (10.12)$$

$$= \frac{1}{2} \sqrt{n+m+1} \delta_{|n-m|,1} \quad (10.13)$$

$$p_{nm} = \langle n | p | m \rangle = -\frac{i}{\sqrt{2}} \sqrt{m+1} \delta_{n,m+1} + \frac{i}{\sqrt{2}} \sqrt{m} \delta_{n,m-1} \quad (10.14)$$

Από την παραπάνω σχέση μπορούμε εύκολα να υπολογίσουμε την Χαμιλτονιανή του αναρμονικού ταλαντωτή

$$H = H_0 + \lambda x^4 \quad (10.15)$$

και τα στοιχεία του πίνακα:

$$H_{nm} \equiv \langle n | H | m \rangle = \langle n | H_0 | m \rangle + \lambda \langle n | x^4 | m \rangle \quad (10.16)$$

$$= (n + \frac{1}{2})\delta_{n,m} + \lambda(x^4)_{nm} \quad (10.17)$$

όπου το $(x^4)_{nm}$ μπορούμε να υπολογίσουμε από τη σχέση (10.12):

$$(x^4)_{nm} = \sum_{i_1, i_2, i_3=0}^{\infty} x_{ni_1} x_{i_1 i_2} x_{i_2 i_3} x_{i_3 m} . \quad (10.18)$$

Το πρόβλημα ανάγεται στον υπολογισμό των ιδιοτιμών του πίνακα H_{nm} .

10.2 ΥΠΟΛΟΓΙΣΜΟΣ ΙΔΙΟΤΙΜΩΝ H_{nm}

Αρχικά επιλέγουμε το μέγεθος της βάσης/πινάκων, έστω N . Χρησιμοποιούμε τις σχέσεις που αναφέραμε στην προηγούμενη παράγραφο για να υπολογίσουμε τους πίνακες x , H_0 , H . Για παράδειγμα, όταν $N = 4$ έχουμε:

$$x = \begin{pmatrix} 0 & \frac{1}{\sqrt{2}} & 0 & 0 \\ \frac{1}{\sqrt{2}} & 0 & 1 & 0 \\ 0 & 1 & 0 & \sqrt{\frac{3}{2}} \\ 0 & 0 & \sqrt{\frac{3}{2}} & 0 \end{pmatrix} \quad (10.19)$$

$$H_0 = \begin{pmatrix} \frac{1}{2} & 0 & 0 & 0 \\ 0 & \frac{3}{2} & 0 & 0 \\ 0 & 0 & \frac{5}{2} & 0 \\ 0 & 0 & 0 & \frac{7}{2} \end{pmatrix} \quad (10.20)$$

$$H = \begin{pmatrix} \frac{1}{2} + \frac{3\lambda}{4} & 0 & \frac{3\lambda}{\sqrt{2}} & 0 \\ 0 & \frac{3}{2} + \frac{15\lambda}{4} & 0 & 3\sqrt{\frac{3}{2}}\lambda \\ \frac{3\lambda}{\sqrt{2}} & 0 & \frac{5}{2} + \frac{27\lambda}{4} & 0 \\ 0 & 3\sqrt{\frac{3}{2}}\lambda & 0 & \frac{7}{2} + \frac{15\lambda}{4} \end{pmatrix} \quad (10.21)$$

Σκοπός μας είναι να γράψουμε ένα πρόγραμμα που θα υπολογίζει τις ιδιοτιμές $E_n(\lambda)$. Για το λόγο αυτό είναι αναγκαίο ο αναγνώστης να ανατρέξει στο 5ο και 6ο κεφάλαιο των σημειώσεων. Αντί όμως να γράψουμε τον δικό μας κώδικα για τον υπολογισμό των ιδιοτιμών και ιδιοδιανυσμάτων των πινάκων που μας ενδιαφέρουν θα χρησιμοποιήσουμε τις έτοιμες ρουτίνες που υπάρχουν

στη βιβλιοθήκη LAPACK. Η βιβλιοθήκη αυτή υπάρχει στον δικτυακό τόπο <http://www.netlib.org/lapack/> και λεπτομέρειες για τη χρήση της μπορούν να βρεθούν στο <http://www.netlib.org/lapack/lug/>. Επισκευήτε το δικτυακό τόπο και αναζητήστε ρουτίνες που σας ενδιαφέρουν.

Ως απλοί άπειροι χρήστες, θα αναζητήσουμε “ρουτίνες οδηγούς” (driver routines) που κάνουν μια δουλειά ολοκληρωμένα. Έχουμε να διαγωνιστούμε πίνακα συμμετρικό και διαλέγουμε την ρουτίνα DSYEV (D = double precision, SY = symmetric, EV = eigenvalues with optional eigenvectors). Οι συναρτήσεις της LAPACK έχουν βοήθεια online από τα man pages του συστήματος (Unix/Linux). Η εντολή που δίνουμε είναι η

```
man dsyev
```

Από εκεί μαθαίνουμε ότι η χρήση της είναι:

```
SUBROUTINE DSYEV( JOBZ, UPLO, N, A, LDA, W, WORK, LWORK, INFO )
    CHARACTER      JOBZ, UPLO
    INTEGER         INFO, LDA, LWORK, N
    DOUBLE          PRECISION A( LDA, * ), W( * ), WORK( * )
```

ARGUMENTS

```

JOBZ      (input) CHARACTER*1
          = 'N': Compute eigenvalues only;
          = 'V': Compute eigenvalues and eigenvectors.

UPLO      (input) CHARACTER*1
          = 'U': Upper triangle of A is stored;
          = 'L': Lower triangle of A is stored.

N         (input) INTEGER
          The order of the matrix A.  N >= 0.

A         (input/output) DOUBLE PRECISION array, dimension (LDA, N)
          On entry, the symmetric matrix A. If UPLO = 'U', the leading
          N-by-N upper triangular part of A contains the upper triangular
          part of the matrix A. If UPLO = 'L', the leading N-by-N lower
          triangular part of A contains the lower triangular part of the
          matrix A. On exit, if JOBZ = 'V', then if INFO = 0, A contains
          the orthonormal eigenvectors of the matrix A. If JOBZ = 'N',
          then on exit the lower triangle (if UPLO='L') or the upper tri-
          angle (if UPLO='U') of A, including the diagonal, is destroyed.

LDA       (input) INTEGER
          The leading dimension of the array A.  LDA >= max(1,N).
```

W (output) DOUBLE PRECISION array, dimension (N)
If INFO = 0, the eigenvalues in ascending order.

WORK (workspace/output) DOUBLE PRECISION array, dimension (LWORK)
On exit, if INFO = 0, WORK(1) returns the optimal LWORK.

LWORK (input) INTEGER
The length of the array WORK. LWORK $\geq \max(1, 3*N-1)$. For optimal efficiency, LWORK $\geq (NB+2)*N$, where NB is the block-size for DSYTRD returned by ILAENV.

If LWORK = -1, then a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued by XERBLA.

INFO (output) INTEGER
= 0: successful exit
< 0: if INFO = -i, the i-th argument had an illegal value
> 0: if INFO = i, the algorithm failed to converge; i off-diagonal elements of an intermediate tridiagonal form did not converge to zero.

Οδηγούμαστε λοιπόν στο να γράψουμε τον εξής κώδικα για να δοκιμάσουμε τη χρήση της σε ένα πίνακα A(N,N):

```

PROGRAM TEST_EVS
IMPLICIT NONE
INTEGER P,LWORK ! P= megisth diastash pinaka
PARAMETER(P=100,LWORK=3*P-1)
DOUBLE PRECISION A(P,P),W(P),WORK(LWORK)

INTEGER N !the dimension of the **used** part of the matrix A(N,N)
INTEGER I,J
CHARACTER *1,JOBZ,UPLO
INTEGER LDA,INFO

C  Orizoume ton **symmetriko** pinaka pou 0a diagwniopoishoume.
C  H rutina xrhsimopoei mono to anw trigwniko meros tou (UPLO='U')
C  opote to katw trigwniko den xreiazetai na oristei.
N=4
A(1,1)=-7.7

```

```

A(1,2)= 2.1
A(1,3)=-3.7
A(1,4)= 4.4
A(2,2)= 8.3
A(2,3)=-16.
A(2,4)= 4.6
A(3,3)=-12.
A(3,4)=-1.04
A(4,4)=-3.7
C   Ton Typwnoume gia elegxo: (prin na kalesoume th rutina giati
C   meta katastrefetai!!)
DO I=1,N
  DO J=i,N
    PRINT *, 'A( ',I,' , ',J,' )=',A(I,J)
  ENDDO
ENDDO
C   Orizoume na ypologistoun kai ta idiodianysmata:
JOBZ='V'
UPLO='U'
PRINT *, 'COMPUTING WITH DSYEV:'
LDA=P                                !notice that LDA-> P>N !!
CALL DSYEV(JOBZ,UPLO,N,A,LDA,W,WORK,LWORK,INFO)
PRINT *, 'DSYEV: DONE. CHECKING NOW:'
C   An to INFO den einai 0 tote exoyme sfalma:
IF(INFO .NE. 0)THEN
  PRINT *, 'DSYEV FAILED. INFO= ',INFO
  STOP
ENDIF
C   Typwnoume ta apotelesmata, W(I) exei tis idiotimes:
PRINT *, 'DSYEV: DONE.: '
PRINT *, 'EIGENVALUES OF MATRIX:'
DO I=1,N
  PRINT *, 'LAMBDA( ',I,' )=',W(I)
ENDDO
C   O pinakas A exei ta idiodianysmata **stis sthles tou**:
C   (giati h fortran "trexei" ton mesa deikth grhgorotera anti0eta
C   apo thn C...)
PRINT *, 'EIGENVECTORS OF MATRIX'
DO J=1,N
  PRINT*, 'EIGENVECTOR ',J,' FOR EIGENVALUE ',W(J)
  DO I=1,N
    PRINT*, 'V_ ',J,' ( ',I,' )= ',A(I,J)
  
```

```

ENDDO
ENDDO
END

```

Το επόμενο βήμα είναι να μεταγλωττίσουμε τον κώδικα. Το σημείο που πρέπει να προσέξουμε είναι ότι στο στάδιο της σύνδεσης (linking) πρέπει να δώσουμε οδηγίες στον loader `ld` που βρίσκονται οι βιβλιοθήκες `LAPACK` και η `BLAS` (η βασικές υπολογιστικές ρουτίνες γραμμικής άλγεβρας είναι στην `BLAS`). Όλες οι συναρτήσεις είναι μεταγλωττισμένες και τα `object files` τους είναι αρχειοθετημένα στα αρχεία `liblapack.a` `libblas.a` που μπορούμε να αναζητήσουμε με τις εντολές:

```

locate libblas
locate liblapack

```

Για να δούμε τα περιεχόμενά τους δίνουμε τις εντολές:

```

ar -t /usr/lib/libblas.a
ar -t /usr/lib/liblapack.a

```

(ή αντικαθιστούμε το `/usr/lib` με τη διαδρομή που αντιστοιχεί στο σύστημά μας). Αν ο κώδικάς μας είναι στο αρχείο `test.f` για τη μεταγλώττιση δίνουμε την εντολή:

```
f77 test.f -o test -L/usr/lib -llapack -lblas
```

Η επιλογή `-L/usr/lib` λέει στον loader να αναζητήσει τις βιβλιοθήκες στο `/usr/lib` (άχρηστο στην περίπτωσή μας γιατί το ψάχνει έτοιμο και αλλιώς, χρήσιμο αν έχουμε βιβλιοθήκες σε μη συμβατικά μέρη) ενώ οι `-llapack -lblas` του λένε να ζητήσει όποια σύμβολα δεν έχουν ξεκαθαριστεί πρώτα στη βιβλιοθήκη `liblapack.a` και μετά στην `libblas.a`

Η παραπάνω εντολή έχει ως αποτέλεσμα το εκτελέσιμο αρχείο `test` που όταν το τρέξουμε παίρνουμε το αποτέλεσμα:

```

EIGENVALUES OF MATRIX:
LAMBDA( 1)= -21.4119907
LAMBDA( 2)= -9.93394359
LAMBDA( 3)= -2.55765591
LAMBDA( 4)=  18.8035905
EIGENVECTORS OF MATRIX
EIGENVECTOR  1 FOR EIGENVALUE  -21.4119907
V_ 1( 1)=  -0.197845668
V_ 1( 2)=  -0.464798676
V_ 1( 3)=  -0.854691009
V_ 1( 4)=   0.119676904
EIGENVECTOR  2 FOR EIGENVALUE  -9.93394359

```

```

V_ 2( 1)=  0.824412399
V_ 2( 2)= -0.132429396
V_ 2( 3)= -0.191076519
V_ 2( 4)= -0.516039161
EIGENVECTOR  3 FOR EIGENVALUE  -2.55765591
V_ 3( 1)=  0.502684215
V_ 3( 2)= -0.247784372
V_ 3( 3)=  0.132853329
V_ 3( 4)=  0.817472616
EIGENVECTOR  4 FOR EIGENVALUE   18.8035905
V_ 4( 1)=  0.168848655
V_ 4( 2)=  0.839659187
V_ 4( 3)= -0.464050682
V_ 4( 4)=  0.226096318

```

Τώρα είμαστε έτοιμοι να λύσουμε το πρόβλημα του αναρμονικού ταλαντωτή. Το πρόγραμμα βρίσκεται στην ιστοσελίδα του μαθήματος www.physics.ntua.gr/ph36/Lectures/15/anharmonic.f από όπου μπορείτε να το κατεβάσετε. Στην κύρια ρουτίνα του προγράμματος ο χρήστης εισάγει τις βασικές παραμέτρους, τη διάσταση του χώρου Hilbert DIM και τις τιμές του λ για τις οποίες επιθυμεί να υπολογιστούν οι ιδιοτιμές του τελεστή $H(\lambda)$. Οι τελευταίες ορίζονται από τις μεταβλητές `lambda0`, `lambdaf`, `dlambda` που αντιστοιχούν στις μέγιστες και ελάχιστες τιμές λ_{min} , λ_{max} και το βήμα $\delta\lambda$ αντίστοιχα (**Ερώτηση**: Στο πρόγραμμα που δίνεται παρακάτω, θα συμπεριλαμβάνεται το λ_{max} στις τιμές που γίνεται ο υπολογισμός ή όχι;). Το πρόγραμμα καλεί την υπορουτίνα `calculate_X4` για να υπολογίσει τον πίνακα $(x^4)_{nm}$ του τελεστή x^4 στην $\{|v\rangle\}$ αναπαράσταση. Ο υπολογισμός στην υπορουτίνα αυτή είναι ακριβή και μπορεί να γίνει πολύ γρηγορότερα υπολογίζοντας εύκολα τα $(x^4)_{nm}$ αναλυτικά. Αυτό αφήνεται σαν άσκηση στον αναγνώστη. Ο υπολογισμός γίνεται μία φορά αφού ο πίνακας είναι αναξάρτητος του λ . Στη συνέχεια για κάθε τιμή του λ υπολογίζονται οι ιδιοτιμές του $H(\lambda)$ καλώντας την υπορουτίνα `calculate_evs` και τα αποτελέσματα τυπώνονται στο `stdout` (standard output).

Η υπορουτίνα `calculate_evs` καλεί την `calculate_H` να υπολογίσει τον πίνακα $H(\lambda)_{nm}$ η οποία κάνει χρήση των σχέσεων (10.16). Στη συνέχεια καλείται η `DSYEV` της LAPACK να κάνει τη διαγωνιοποίηση. Προσέχουμε στο όριο LDA της `DSYEV` να βάλουμε τη σωστή διάσταση του πίνακα H που είναι P και όχι DIM. Στη συνέχεια παρατίθεται ο κώδικας:

```

program anharmonic_elevels
implicit none
integer P,LWORK
parameter(P=1000,LWORK=3*P-1)
double precision H (P,P),X(P,P) !teletes: Hamiltonian,Position
double precision X4(P,P)         !telesths: X^4

```



```

double precision E(P)          !energeiakies idiotimes
double precision WORK(LWORK)   !bohOhtikos xwros ergasias LAPACK
double precision lambda,lambda0,lambdaf,dlambda
integer DIM
integer i

C      O xrhsth kaOorizei to megeOos ths bashs:
print *, 'Enter Hilbert Space dimension:'
read(5,*)DIM
C      O Xrhsth dinei elaxisth/megisth timh sto lambda kai bhma
C      ypologismou:
print *, 'Enter lambda0,lambdaf,dlambda:'
read(5,*)lambda0,lambdaf,dlambda
print *, 'lambda0= ',lambda0
C      Print Message:
print *, '# #####'
print *, '# Calculation of energy levels of anharmonic oscillator'
print *, '# using matrix methods.'
print *, '# Hilbert Space Dimension = ',DIM
print *, '# lambda coupling = ',lambda0,' - ',lambdaf,
*      ' step= ',dlambda
print *, '# #####'
print *, '# Output: lambda E_0 E_1 .... E_{N-1}'
print *, '# -----'

C      Ypologizoume ton telesth X^4:
call calculate_X4(X,X4,DIM)
C      Kai twra tis idiotimes synarthsei toy lambda:
do lambda=lambda0,lambdaf,dlambda
    call calculate_evs(H,X4,E,WORK,lambda,DIM)
C      Prosekte to format gia na mhn typwnontai oi idiotimes se
C      diaforetikh grammh an to N einai megalο
    write(6,100)'EV ',lambda,(E(i),i=1,DIM)
enddo
100 FORMAT(A3,100G25.15)
end

subroutine calculate_evs(H,X4,E,WORK,lambda,DIM)
implicit none
integer P,LWORK
parameter(P=1000,LWORK=3*P-1)

```

```

double precision H(P,P)      !telestes: Hamiltonian,Position
double precision X4(P,P)     !telesths: X^4
double precision E(P)        !energeiakes idiotimes
double precision WORK(LWORK) !bohOhtikos xwros ergasias LAPACK
integer DIM
double precision lambda

character *1,JOBZ,UPLO
integer LDA,INFO,i,j

call calculate_H(H,X4,lambda,DIM)
JOBZ='N'
UPLO='U'
call dsyev(JOBZ,UPLO,DIM,H,P,E,WORK,LWORK,INFO)
print *,'# ***** EVEC *****'
do j=1,DIM
  write(6,101)'# EVEC ',lambda,(H(i,j), i=1,DIM)
enddo
print *,'# ***** EVEC *****'
101 FORMAT(A7,F6.4,1000G14.6)

C  An to INFO den einai 0 tote exoyme sfalma:
if(INFO .ne. 0)then
  print *,'dsyev failed. INFO= ',INFO
  stop
endif

end

subroutine calculate_H(H,X4,lambda,DIM)
implicit none
integer P,LWORK
parameter(P=1000,LWORK=3*P-1)
double precision H(P,P)      !telestes: Hamiltonian,Position
double precision X4(P,P)     !telesths: X^4
integer DIM
double precision lambda

integer i,j,n,m

do j=1,DIM
  do i=1,DIM

```

```

        H(i,j)=lambda*X4(i,j)
    enddo
    H(j,j) = H(j,j) + DBLE(j) - 0.5D0 !E_n=n+1/2,n=j-1 => E_n=j-1/2
enddo

print *,'# ***** H *****'
do j=1,DIM
    write(6,102)'# HH ',(H(i,j), i=1,DIM)
enddo
print *,'# ***** H *****'

102 FORMAT(A5,1000G14.6)
end

subroutine calculate_X4(X,X4,DIM)
implicit none
integer P
parameter(P=1000)
double precision X(P,P),X4(P,P)      !telesth: X, X^4
integer DIM

integer i,j,m,n,i1,i2,i3
double precision isqrt2 ! 1/sqrt(2)
parameter(isqrt2=0.70710678118654752440D0)
C Arxika ypologizoume ton telesth Oeshs:
do j=1,DIM
    do i=1,DIM
        X (i,j)=0.0D0
        X4(i,j)=0.0D0
    enddo
enddo

C Kai twra ta mh mhdenika stoixeia:
do i=1,DIM
    n=i-1 !deiktes 0,...,DIM-1
C O oros delta_{n,m+1}, dhladh m=n-1
    m=n-1 !to energeiako epipedo n -> i=n+1, m-> j=m+1
    j=m+1
    X(i,j)=isqrt2*dsqrt(DBLE(m+1))
C O oros delta_{n,m-1}, dhladh m=n+1
    m=n+1
    j=m+1
    X(i,j)=isqrt2*dsqrt(DBLE(m))

```

```


enddo

C   Kai twra ypologizoume ton telesth H:
C   Arxizoume me ton oro X^4:
do j=1,DIM
  do i=1,DIM
    do i1=1,DIM
      do i2=1,DIM
        do i3=1,DIM
          X4(i,j)=X4(i,j)+X(i,i1)*X(i1,i2)*X(i2,i3)*X(i3,j)
        enddo
      enddo
    enddo
  enddo
enddo

C   Parathrhsh: Xanoume adiko xrono gia ton ypologismo toy X4 giati
C   ta perissotera stoixeia tou einai 0. Efoson o analytikos ypologismos
C   einai eykolos, systhnetai na ginei kai na ypologizontai mono ta
C   mh mhdenika stoixeia!! (na ginei san askhsh)
end

```


Στη συνέχεια παρουσιάζουμε ένα υπολογισμό για $N = 10, 20, 30$. Στο σχήμα φαίνεται καλά ότι οι 3 πρώτες ενεργειακές στάθμες έχουν συγκλίνει καλά για τις τιμές αυτές. Επίσης γίνεται φανερό ότι αποκλίσεις παρουσιάζονται για μεγάλες τιμές του λ και του N .



Σχήμα 10.1: Οι πρώτες 12 ενεργειακές στάθμες του αναρμονικού ταλαντωτή για τιμές του $0 \leq \lambda < 1$ για $N = 10, 20, 30$.

Η σύγκλιση μίας συγκεκριμένης ιδιοτιμής E_n καθώς $N \rightarrow \infty$ για κάποια

τιμή του λ φαίνεται στα παρακάτω σχήματα όπου φαίνεται ότι η σύγκλιση είναι γρήγορη για μικρά λ και η ενώ αργή για μεγάλα.



Σχήμα 10.2: Η ενεργειακή στάθμη E_0 για $\lambda = 0.2, 0.9$ σαν συνάρτηση της (αντίστροφης) διάστασης του χώρου Hilbert $1/N$. Σύγκλιση επιτυγχάνεται για μικρές τιμές του N ενώ φαίνεται ότι για $\lambda=0.2$ γίνεται ελαφρώς γρηγορότερα από ότι για $\lambda=0.9$.

10.3 ΤΟ ΔΙΠΛΟ ΠΗΓΑΔΙ ΔΥΝΑΜΙΚΟΥ

θα χρησιμοποιήσουμε τις μεθόδους πινάκων που αναφέραμε για να υπολογίσουμε τα ενεργειακά επίπεδα σωματιδίου μέσα στο διπλό πηγάδι δυναμικού. Αυτό δίνεται από τη Χαμιλτονιανή:

$$H = \frac{p^2}{2} - \frac{x^2}{2} + \lambda \frac{x^4}{4} \quad (10.22)$$


και τα σημεία ισορροπίας στην κλασσική κίνηση βρίσκονται στα ελάχιστα:

$$x_0 = \pm \frac{1}{\sqrt{\lambda}}, \quad V_{min} = -\frac{1}{4\lambda} \quad (10.23)$$

Όταν το πηγάδι είναι πολύ βαθύ τότε για τις χαμηλότερες στάθμες μπορούμε να θεωρήσουμε ότι το δυναμικό προσεγγίζεται καλά από αυτό του αρμονικού ταλαντωτή με συχνότητα $\omega^2 = V''(x_0)$ οπότε

$$E_{min} \approx V_{min} + \frac{1}{2}\omega \quad (10.24)$$

Στην περίπτωση αυτή το φαινόμενο σύραγγας είναι πολύ ασθενές με αποτέλεσμα τα ενεργειακά επίπεδα να χωρίζονται ελαφρώς μεταξύ τους ανά ζεύγη. Αυτό γίνεται γιατί οι αντίστοιχες ιδιοκαταστάσεις είναι συμμετρικοί και αντισυμμετρικοί



Σχήμα 10.3: Η ενεργειακή στάθμη E_9 για $\lambda = 0.2, 0.9$ σαν συνάρτηση της (αντίστροφης) διάστασης του χώρου Hilbert $1/N$.

συνδυασμοί κυματοσυναρτήσεων που αντιστοιχούν σε καταστάσεις εντοπισμένες στο αριστερό ή δεξιό ελάχιστο της δυναμικής ενέργειας. Π.χ. για τα δύο χαμηλότερα ενεργειακά επίπεδα περιμένουμε ότι

$$E_{0,1} = E_{min} \pm \frac{\Delta}{2} \quad (10.25)$$


όπου $\Delta \ll |E_{min}|$.

Ως βάση για τον υπολογισμό της Χαμιλτονιανής (10.22) θα χρησιμοποιήσουμε τις σχέσεις (10.12). Οι απαραίτητες μεταβολές στον κώδικα μας είναι ελάχιστες. Απλά θα προσθέσουμε μία ρουτίνα που να υπολογίζει τους πίνακες p_{nm} . Παίρνουμε έτσι τον κώδικα που αποθηκεύουμε στο αρχείο doublewell.f:

```

program doublewell_elevels
implicit none
integer P,LWORK
parameter(P=1000,LWORK=3*P-1)
double precision H (P,P),X(P,P) !telestes: Hamiltonian,Position
double precision X4(P,P)         !telesths: X^4
double precision X2(P,P)         !telesths: X^2
double precision iP(P,P),P2(P,P)!telestes: i P, P^2
double precision H0(P,P)         !telesths: H_0=1/2 P^2-1/2 X^2
double precision E(P)            !energeiakas idiotimes
double precision WORK(LWORK)     !bohOhtikos xwros ergasias LAPACK
double precision lambda,lambda0,lambdaf,dlambda
integer DIM0,DIMF,dDIM,DIM
integer i

```




Σχήμα 10.4: Η ενεργειακή στάθμη E_{20} για $\lambda = 0.2, 0.9$ σαν συνάρτηση της (αντίστροφης) διάστασης του χώρου Hilbert $1/N$. Η σύγκλιση δεν έχει επιτευχθεί για τις προβαλλόμενες τιμές του N με την ακρίβεια των παραπάνω σχημάτων.

```

C      O xrhsthhs ka0orizei to mege0os ths bashs:
      print *, 'Enter Hilbert Space dimensions (DIM0,DIMF,DDIM):'
      read(5,*)DIM0,DIMF,DDIM
C      O Xrhsthhs dinei elaxisth/megisth timh sto lambda kai bhma
C      ypologismou:
      print *, 'Enter lambda0,lambdaf,dlambda:'
      read(5,*)lambda0,lambdaf,dlambda
      print *, 'lambda0= ',lambda0
C      Print Message:
      print *, '# #####'
      print *, '# Calculation of energy levels of double well potential'
      print *, '# using matrix methods.'
      print *, '# Hilbert Space Dimensions = ',DIM0,' - ',DIMF,
$      ' step= ',dDIM
      print *, '# lambda coupling = ',lambda0,' - ',lambdaf,
$      ' step= ',dlambda
      print *, '# #####'
      print *, '# Output: DIM lambda E_0 E_1 ... E_{N-1}'
      print *, '# -----'

      do DIM=DIM0,DIMF,dDIM
C      Ypologizoume telestes:
      call calculate_operators(X,X2,X4,iP,P2,H0,DIM)
C      Ypologizoume thn Hamiltonian H_0 xwris oro X^4

```



Σχήμα 10.5: Η δυναμική ενέργεια $V(x)$ για $\lambda = 0.1, 0.2$.

```

C      Kai twra tis idiotimes synarthsei toy lambda:
      do lambda=lambda0,lambdaf,dlambda
        call calculate_evs(H,H0,X4,E,WORK,lambda,DIM)
C      Prosexe to format gia na mhn typwontai oi idiotimes se
C      diaforetikh grammh an to N einai megalo
        write(6,100)'EV ',DIM,lambda,(E(i),i=1,DIM)
      enddo
    enddo
100  FORMAT(A3,I5,1000G25.15)
    end

subroutine calculate_evs(H,H0,X4,E,WORK,lambda,DIM)
implicit none
integer P,LWORK
parameter(P=1000,LWORK=3*P-1)
double precision H(P,P)      !telestes: Hamiltonian,Position
double precision X4(P,P)      !telesths: X^4
double precision H0(P,P)      !telesths: H_0=1/2 P^2-1/2 X^2
double precision E(P)         !energeiakes idiotimes
double precision WORK(LWORK)  !bohOhtikos xwros ergasias LAPACK
integer DIM
double precision lambda

character *1,JOBZ,UPLD
integer LDA,INFO,i,j

```



```

call calculate_H(H,H0,X4,lambda,DIM)
JOBZ='V'
UPLO='U'
call dsyev(JOBZ,UPLO,DIM,H,P,E,WORK,LWORK,INFO)
print *,'# ***** EVEC *****'
do j=1,DIM
  write(6,101)'# EVEC ',DIM,lambda,(H(i,j), i=1,DIM)
enddo
print *,'# ***** EVEC *****'
101 FORMAT(A7,I5,F8.4,1000G14.6)

C   An to INFO den einai 0 tote exoyme sfalma:
if(INFO .ne. 0)then
  print *,'dsyev failed. INFO= ',INFO
  stop
endif

end

subroutine calculate_H(H,H0,X4,lambda,DIM)
implicit none
integer P,LWORK
parameter(P=1000,LWORK=3*P-1)
double precision H(P,P)      !telestes: Hamiltonian,Position
double precision H0(P,P)      !telesths:  $H_0=1/2 P^2-1/2 X^2$ 
double precision X4(P,P)      !telesths:  $X^4$ 
integer DIM
double precision lambda

integer i,j,n,m

do j=1,DIM
  do i=1,DIM
    H(i,j)=H0(i,j)+0.25D0*lambda*X4(i,j)
  enddo
enddo

print *,'# ***** H *****'
do j=1,DIM
  write(6,102)'# HH ',(H(i,j), i=1,DIM)
enddo
print *,'# ***** H *****'

```

```

102 FORMAT(A5,1000G14.6)
end

subroutine calculate_operators(X,X2,X4,iP,P2,H0,DIM)
implicit none
integer P
parameter(P=1000)
double precision X(P,P),X4(P,P)      !telesths: X, X^4
double precision X2(P,P)              !telesths: X^2
double precision iP(P,P),P2(P,P)!telestes: i P, P^2
double precision H0(P,P)              !telesths: H_0=1/2 P^2-1/2 X^2
integer DIM

integer i,j,m,n,i1
double precision isqrt2 ! 1/sqrt(2)
parameter(isqrt2=0.70710678118654752440D0)
C Arxika ypologizoume ton telesth 0eshs,i*ormhs:
do j=1,DIM
do i=1,DIM
X (i,j)=0.0D0
X4(i,j)=0.0D0
X2(i,j)=0.0D0
iP(i,j)=0.0D0
P2(i,j)=0.0D0
enddo
enddo
C Kai twra ta mh mhdenika stoixeia:
do i=1,DIM
n=i-1 !deiktes 0,...,DIM-1
C 0 oros delta_{n,m+1}, dhladh m=n-1
m=n-1 !to energeiako epipedo n -> i=n+1, m-> j=m+1
j=m+1
X (i,j) = isqrt2*dsqrt(DBLE(m+1))
iP(i,j) = isqrt2*dsqrt(DBLE(m+1))
C 0 oros delta_{n,m-1}, dhladh m=n+1
m=n+1
j=m+1
X (i,j) = isqrt2*dsqrt(DBLE(m))
iP(i,j) = -isqrt2*dsqrt(DBLE(m))
enddo

```

```

do j=1,DIM
do i=1,DIM
do i1=1,DIM
X2(i,j)=X2(i,j)+ X(i,i1)* X(i1,j)
P2(i,j)=P2(i,j)-iP(i,i1)*iP(i1,j)
enddo
enddo
enddo
C o oros X^4:
do j=1,DIM
do i=1,DIM
do i1=1,DIM
X4(i,j)=X4(i,j)+X2(i,i1)*X2(i1,j)
enddo
enddo
enddo
C kai h hamiltonianh:
do j=1,DIM
do i=1,DIM
H0(i,j)=0.5D0*( P2(i,j)-X2(i,j) )
enddo
enddo

C Parathrhsh: Xanoume adiko xrono gia ton ypologismo toy X4,P4 giati
C ta perissotera stoixeia tou einai 0. Efoson o analytikos ypologismos
C einai eykolos, systhnetai na ginei kai na ypologizontai mono ta
C mh mhdenika stoixeia!! (na ginei san askhsh)
end

```

10.4 ΑΣΚΗΣΕΙΣ

- Υπολογίστε αναλυτικά τον πίνακα $H(\lambda)$ για $N = 2, 3$. Υπολογίστε τις ιδιοτιμές για $N = 2$. Συγκρίνετε με τις τιμές που υπολογίζει το πρόγραμμά σας ως επιβεβαίωση ότι τρέχει σωστά.
- Αλλάξτε τη σειρά -llapack -lblas σε -lblas -llapack στην εντολή μεταγλώττισης. Τι παρατηρήτε; Γιατί;
- Μεταβάλλετε τον κώδικα test.f έτσι ώστε να επιβεβαιώνει ότι τα ιδιοδιανύσματα ικανοποιούν τις σχέσεις $\mathbf{A} \mathbf{v}_i = \lambda_i \mathbf{v}_i$ και ότι αποτελούν ορθοκανονική βάση $\mathbf{v}_i \cdot \mathbf{v}_j = \delta_{ij}$.
- Υπολογίστε τον πίνακα του τελεστή x^4 αναλυτικά και προγραμματίστε τον στο πρόγραμμά σας. Συγκρίνετε τους χρόνους που τρέχει το πρόγραμμά

σας σε σχέση με πριν σαν συνάρτηση του N . Τι συμπεραίνετε;